CSE 505 - Final Project Report

A Physician Advisory System for Chronic Heart Failure management based on knowledge patterns

ZHUO CHEN, KYLE MARPLE, ELMER SALAZAR, GOPAL GUPTA and LAKSHMAN TAMIL

University of Texas at Dallas, Texas, U.S.A

1 Introduction

Management of chronic diseases such as chronic heart failure (CHF) is a major problem in health care. A standard approach followed by the medical community is to have a committee of experts develop guidelines that all physicians should follow. These guidelines typically consist of a series of complex rules that make recommendations based on a patient's information. Due to their complexity, often the guidelines are ignored or not complied with at all. It is not even clear whether it is humanly possible to follow these guidelines due to their length and complexity.

In this paper the author describes a physician-advisory system for CHF management that codes the entire set of clinical practice guidelines for CHF using answer set programming (ASP). Their approach is based on developing reasoning templates, that we call knowledge patterns, and using them to systemically code the clinical guidelines for CHF as ASP rules. Given a patient's medical information, our system generates a recommendation for treatment just as a human physician would, using the guidelines. [1]

2 Methodology

The physician-advisory system for CHF management has two major components, the rule database and the fact table.

2.1 Creating Rule Database

The rule database covers all the knowledge in 2013 ACCF/AHA Guideline for the Management of Heart Failure (Yancy et al. 2013). There are some sixty odd rules in the 2013 ACCF/AHA Guideline for the Management of Heart Failure. The ACCF/AHA guidelines are written in English and are quite complex. These guidelines had to be coded in ASP. In order to do this, we used certain reasoning templates that are called knowledge patterns. These knowledge patterns are quite general and serve as solid building blocks for systematically translating the specifications written in English to ASP.

"In all patients with a recent or remote history of MI or ACS and reduced EF, evidence-based beta blockers should be used to reduce mortality."

In s(ASP), this rule will be coded as shown below. Note that Prolog conventions are followed (variables begin with an upper case letter).

```
recommendation(beta_blockers, class_1):- accf_stage(b),
    history_of_mi_or_acs, measurement(lvef, Data),
    reduced_ef(Data), not contraindication(beta_blockers).
```

Fig. 1. ACCF/AHA rule to Prolog

2.2 Generating Fact Table

The fact table contains the relevant information of the patient with heart failure. The fact table is derived from a patient's electronic health record and from a telemedicine system used to measure vital signs. The patient information consists mainly of: 5 pieces of demographics information, 8 measurements and 25 types of HF-related diseases and symptoms.

Table 1. *Input of the Physician-Advisory System for CHF Management*

Demographics	Gender; age; race
Measurements	Weight; creatinine; potassium; sinus rhythm; left bundle branch block; non-left bundle branch block; QRS duration; ejection fraction NYHA class; ACCF/AHA stage;
Diseases and Symptoms	Sleep apnea, acute coronary syndrome; myocardial infarction; obesity; diabetes; stroke; fluid retention; angioedema; ischemic attack; thromboembolism; elevated plasma natriuretic peptide level; asymptomatic ischemic cardiomyopathy; lipid disorders; hypertension; atrial fibrillation; myocardial ischemia; coronary artery disease; dilated cardiomyopathy; acute profound hemodynamic compromise; threatened end organ dysfunction; ischemic heart disease;angina; structural cardiac abnormalities; atrioventricular block; volume overload
Miscellany	Expectation of survival;pregnancy; history of cardiovascular hospitalization; history of standard neurohumoral antagonist therapy; risk of cardioembolic stroke; eligibility of significant ventricular pacing; eligibility of mechanical circulatory support; dependence of continuous parenteral inotropic; ischemic etiology of HF; requirement of ventricular pacing

Fig. 2. Various Patient Inputs

2.3 Obtaining Output Recommendation

Treatment recommendations returned by the system may include: 11 pharmaceutical treatments, 9 management objectives, and 4 device/surgery therapies.

Table 2. *Output of the Physician-Advisory System for CHF Management*

Pharmaceutical Treatments	ACE inhibitors; ARBs; Beta blockers; statin; diuretics; aldosterone receptor antagonists; hydralazine and isosorbide dinitrate; digoxin; anticoagulations; Omege-3 fatty acids; inotropes;
Management Objectives	Systolic blood pressure control; diastolic blood pressure control; obesity control; diabetes control; tobacco avoidance; cardiotoxic agents avoidance; atrial fibrillation control; water restriction; sodium restriction;
Device/Surgery Therapies	Implantable cardioverter-defibrillator; cardiac resynchronization therapy; mechanical circulatory support; coronary revascularization

Fig. 3. Output Recommendations

3 Implementation

Implementation of the system is based on identifying knowledge patterns and using them as building blocks for constructing the ASP(Answer Set Programming) code.

3.1 s(ASP) system

We run our system on top of the s(ASP) system, a goal-directed, predicate ASP system that can be thought of as Prolog extended with negation based on the stable model semantics [2]. s(ASP) is an implementation of the stable model semantics of logic programming i.e., logic programs extended with negation, in the presence of predicates with arbitrary terms. Such programs need not have a finite grounding, so traditional methods do not apply. Using this method, a normal logic program with predicates can be executed directly under the stable model semantics without requiring it to be grounded either before or during execution and without requiring that its variables range over a finite domain.

Steps to configure s(ASP) system:

1. Building s(ASP) requires **SWI Prolog** to be installed on the local machine.

2. Compile s(ASP) to generate an executable file. Locate src/folder and enter the following command:
   swipl -L0 -G0 -T0 –goal=main –stand_alone=true -o sasp -c main.pl
3. Write Program with Rules and Facts.
4. Run Program: ***sasp -i advisory_system.lp***

3.2 Obtain Rules from the 2013 ACCF/AHA Guidelines

There are some sixty odd rules in the 2013 ACCF/AHA Guideline for the Management of Heart Failure. All of these rules are coded in ASP to run on the s(ASP) system. While developing these knowledge patterns and coding them in ASP, certain facts had to be noted:

1. Multiple rules can lead to the recommendation of a treatment.
2. Multiple rules can lead to contraindication of a treatment.
3. A treatment cannot be recommended if at least one contraindication for that treatment is present.
4. A given treatment recommendation can impact the recommendation and/or contraindication of other treatments.

The knowledge patterns we used are:

Aggressive Reasoning: Take an action (e.g., recommend treatment) if there is a reason. No evidence of danger means there is no danger in taking that action.

```
recommendation(Choice) :- preconditions(Choice),
    not contraindication(Choice).
contraindication(Choice) :- dangers(Choice).
```

Fig. 4. Aggressive Reasoning - Template

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
Guideline: "Digoxin can be beneficial in patients with HFrEF, unless
contraindicated, to decrease hospitalizations for HF."
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

recommendation(digoxin, class_2a) :-
    not contraindication(digoxin),
    accf_stage(c), hf_with_reduced_ef.
contraindication(digoxin) :-
    evidence(atrioventricular_block).
```

Fig. 5. Aggressive Reasoning - Example

Conservative Reasoning: A reason for a recommendation is not enough. Evidence that the recommendation is not harmful must be available.

```
recommendation(Choice) :- preconditions(Choice),
    not contraindication(Choice).
contraindication(Choice) :- not -dangers(Choice).
```

Fig. 6. Conservative Reasoning - Template

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
Guideline: "In patients with structural cardiac abnormalities,
including LV hypertrophy, in the absence of a history of MI or ACS,
blood pressure should be controlled in accordance with clinical
practice guidelines for hypertension to prevent symptomatic HF."
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

recommendation(blood_pressure_control, class_1):-
    accf_stage(b),
    diagnosis(structural_cardiac_abnormalities),
    not contraindication(blood_pressure_control).
contraindication(blood_pressure_control):- not -history(mi).
contraindication(blood_pressure_control):- not -history(acs).
```

Fig. 7. Conservative Reasoning - Example

Anti-Recommendation: A choice can be prohibited if evidence of danger can be found.

```
contraindication(choice) :- dangers(Choice).
```

Fig. 8. Anti-Recommendation - Template

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
Guideline: "Anticoagulation is not recommended in patients with
chronic HFrEF without AF, a prior thromboembolic event,
or a cardioembolic source."
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

contraindication(anticoagulation) :-
    not cardioembolic_source,
    not diagnosis(af),
    not history(thromboembolism),
    hf_with_reduced_ef.
```

Fig. 9. Anti-Recommendation - Example

Preference: Use the second-line choice when the first-line choice is not available.

```
recommendation(First_choice) :- conditions_for_both_choices,
    not contraindication(First_choice).
recommendation(Second_choice) :- conditions_for_both_choices,
    contraindication(First_choice),
    not contraindication(Second_choice).
```

Fig. 10. Preference - Template

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
Guideline: "ARBs are recommended in patients with HFrEF with current
or prior symptoms who are ACE inhibitor intolerant, unless
contraindicated, to reduce morbidity and mortality."
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

recommendation(ace_inhibitors, class_1) :-
    not contraindication(ace_inhibitors),
    accf_stage(c), hf_with_reduced_ef.
recommendation(arbs, class_1) :-
    contraindication(ace_inhibitors),
    not contraindication(arbs),
    not taboo_choice(arbs),
    accf_stage(c),
    hf_with_reduced_ef.
```

Fig. 11. Preference - Example

Concomitant Choice: If a choice is made, some other choices are automatically in effect unless they are prohibited.

```
recommendation(Trigger_choice) :- preconditions(Trigger_choice),
    not contraindication(Trigger_choice),
    not skip_concomitant_choice(Trigger_choice).
skip_concomitant_choice(Trigger_choice) :-
    not recommendation(Concomitant_choice),
    not contraindication(Concomitant_choice).
recommendation(Concomitant_choice) :-
    recommendation(Trigger_choice),
    not contraindication(Concomitant_choice).
```

Fig. 12. Concomitant Choice - Template

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
Guideline: "Diuretics should generally be combined with an ACE inhib-
itor, beta blocker, and aldosterone antagonist. Few patients with HF
will be able to maintain target weight without the use of diuretics."
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

recommendation(ace_inhibitors, class_1) :-
    accf_stage(c),
    not skip_concomitant_choice(ace_inhibitors),
    not contraindication(ace_inhibitors),
    hf_with_reduced_ef.
skip_concomitant_treatment(ace_inhibitors) :-
    hf_with_reduced_ef,
    not recommendation(diuretics, class_1),
    not contraindication(diuretics).
recommendation(diuretics, class_1) :-
    hf_with_reduced_ef,
    not contraindication(diuretics),
    recommendation(ace_inhibitors, class_1).
```

Fig. 13. Concomitant Choice - Example

Indispensable Choice: If a choice is made, some other choices must also be made. If those choices can't be made, then the first choice is revoked. Note: Choosing "Trigger choice" forces "Indispensable choice.

```
recommendation(beta_blockers, class_1) :-
    not skip_concomitant_choice(beta_blockers),
    not absent_indispensable_choice(beta_blockers),
    not contraindication(beta_blockers), accf_stage(c), hf_with_reduced_ef.
absent_indispensable_choice(beta_blockers) :-
    not recommendation(diuretics, class_1), hf_with_reduced_ef,
    accf_stage(c), current_or_recent_history_of_fluid_retention.
recommendation(diuretics, class_1) :-
    recommendation(beta_blockers, class_1),
    not contraindication(diuretics), accf_stage(c), hf_with_reduced_ef,
    current_or_recent_history_of_fluid_retention.
```

Fig. 14. Indispensable Choice - Template

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
Guideline: "In patients with a current or recent history of fluid
retention, beta blockers should not be prescribed without diuretics."
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

recommendation(beta_blockers, class_1) :-
    not skip_concomitant_choice(beta_blockers),
    not absent_indispensable_choice(beta_blockers),
    not contraindication(beta_blockers),
    accf_stage(c),
    hf_with_reduced_ef.
absent_indispensable_choice(beta_blockers) :-
    not recommendation(diuretics, class_1),
    hf_with_reduced_ef,
    accf_stage(c),
    current_or_recent_history_of_fluid_retention.
recommendation(diuretics, class_1) :-
    recommendation(beta_blockers, class_1),
    not contraindication(diuretics),
    accf_stage(c),
    hf_with_reduced_ef,
    current_or_recent_history_of_fluid_retention.
```

Fig. 15. Indispensable Choice - Example

Incompatible Choice: Some choices cannot be in effect at the same time.

```
taboo_choice(Choice_1) :-
    recommendation(Choice_2),
    ...,
    recommendation(Choice_n).
taboo_choice(Choice_2) :-
    recommendation(Choice_1),
    recommendation(Choice_3),
    ....
    recommendation(Choice_n).
    ...
taboo_choice(Choice_n) :-
    recommendation(Choice_1),
    recommendation(Choice_2),
    ....
    recommendation(Choice_n-1).
```

```
recommendation(Choice_1) :-
    conditions_for_choice_1,
    not contraindication(Choice_1),
    not taboo_choice(Choice_1).
recommendation(Choice_2) :-
    conditions_for_choice_2,
    not contraindication(Choice_2),
    not taboo_choice(Choice_2).

    ...
recommendation(Choice_n) :-
    conditions_for_choice_n,
    not contraindication(Choice_n),
    not taboo_choice(Choice_n).
```

Fig. 16. Incompatible Choice - Template

```
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
Guideline: "Routine combined use of an ACE inhibitor, ARB, and
aldosterone antagonist is potentially harmful for patients with HFrEF."
%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

taboo_choice(ace_inhibitors) :-
    hf_with_reduced_ef,
    recommendation(arbs, class_1),
    recommendation(aldosterone_antagonist, class_1).
taboo_choice(arbs) :-
    hf_with_reduced_ef,
    recommendation(ace_inhibitors, class_1),
    recommendation(aldosterone_antagonist, class_1).
taboo_choice(aldosterone_antagonist) :-
    hf_with_reduced_ef,
    recommendation(arbs, class_1),
    recommendation(ace_inhibitors, class_1).
recommendation(ace_inhibitors, class_1) :-
    accf_stage(c),
    hf_with_reduced_ef,
    not skip_concomitant_choice(ace_inhibitors),
    not taboo_choice(ace_inhibitors),
    not contraindication(ace_inhibitors).
recommendation(arbs, class_1) :-
    contraindication(ace_inhibitors),
    not contraindication(arbs),
    not taboo_choice(arbs),
    accf_stage(c),
    hf_with_reduced_ef.
recommendation(aldosterone_antagonist, class_1) :-
    conditions_for_aldosterone_antagonist_class_1,
    not skip_concomitant_choice(aldosterone_antagonist),
    not contraindication(aldosterone_antagonist),
    not taboo_choice(aldosterone_antagonist).
```

Fig. 17. Incompatible Choice - Example

3.3 Encode Patient Information as Input

The input to the system is a patient's information, including demographics, history, daily symptoms, risks and measurements, as well as ACCF/AHA stage and NYHA class. When queried for a treatment recommendation, our system is able to give recommendations according to the guideline just as a physician would. The system takes its input from:

1. A patient's electronic health record that includes demographic information, test results, etc.
2. A telemedicine system that provides data about vital signs (heart rate, blood pressure, weight, etc.

Obtain various patient records and pass it as input to the Physician Advisory System for CHF Management. Test the output obtained from the system.

```
% Doctor's Assesement
accf_stage(c).
nyha_class(3).
expectation_of_survival(3).

% Demographics of the patient
gender(female).
age(78).

% Measurements from the Lab
hf_with_reduced_ef.
measurement(creatinine, 1.8).
measurement(potassium, 4.9).
measurement(lvef, 0.35).
measurement(lbbb, 180).
measurement(sinus_rhythm).

% History of the Patient
diagnosis(myocardial_ischemia).
diagnosis(atrial_fibrillation).
diagnosis(coronary_artery_disease).
diagonosis(hypertension).
evidence(ischemic_etiology_of_hf).
evidence(sleep_apnea).
evidence(fluid_retention).
history(mi, recent).
history(stroke).
history(cardiovascular_hospitalization).
post_mi(40).
```

Fig. 18. Patient Input given in Prolog

3.4 Obtain Recommendation as Output

Each treatment recommendation (represented as a partial answer set) contains all of the predicates that must hold in order for the query to be successful.

Run the following query to get recommendations:
recommendation(Treatment, Class)

1. **Interactive Query** Here, the query can be passed through the command prompt.



Fig. 19. Interactive Query Command with Output

2. **Auto Query** Here, the query is mentioned in the code as given below. Number of answer sets to be printed out of the multiple recommendations returned is mentioned besides 'compute'.

```
#compute 3 {recommendation(Treatment,Class)}.
```

Fig. 20. Code to Compute Query



Fig. 21. Auto Query Command with Output

4 Code

The code is uploaded on GitHub. The link to the repository is as given below:
`https://github.com/melitasaldanha/CSE505---Computing-with-Logic`

5 References

[1] ZHUO CHEN, KYLE MARPLE, ELMER SALAZAR, GOPAL GUPTA and LAKSHMAN TAMIL.
A Physician Advisory System for Chronic Heart Failure management based on knowledge patterns.

[2] MARPLE, K., SALAZAR, E. AND GUPTA, G. 2016.
s(ASP). `https://sourceforge.net/projects/saspsystem/`.